# Microcomputer Tools For Communications Engineering




**25 Computer Programs in the Basic Language**

# Microcomputer Tools For Communications Engineering

by

Shing Ted Li
John W. Rockway
James C. Logan
Daniel W. S. Tam

## DISCLAIMER OF WARRANTIES AND LIMITATION OF LIABILITIES

The authors have made every effort to ensure accuracy and completeness in the preparation of this book, and in the development and testing of the computer programs. However, the authors and publisher make no expressed or implied warranty or guarantee of any kind with respect to the completeness, effectiveness, and accuracy of the programs and documentation in this book. In no event will the authors or publisher be liable for incidental or otherwise consequential damages in connection with the furnishing of, or the performance of, or as a result of the use of any of these programs.



**ARTECH HOUSE, INC.**
**610 WASHINGTON ST.**
**DEDHAM, MA 02026**



**International Standard Book Number: 0-89006-132-7**
**Library of Congress Catalog Card Number: 83-071834**

# Microcomputer Tools For Communications Engineering

# Preface

Occasionally, a new technology is developed which is destined to produce significant changes in the way people think and conduct their business. For many decades, scientists and engineers have struggled with unmanageable equations and data using trial and error techniques, employing logarithmic tables and inadequate slide rule calculations. Then came the advent of the digital computer.

In the 1950s and 1960s, physically large and expensive computing machines (that were relatively slow with limited capability compared to today's standards) became available to a few. At first, stored programs were accessible through direct connection of individual terminals at a short distance away. The revolution had begun.

In the 1970s, technologists rushed to convert proven algorithms into computer programs or develop new algorithms useful for efficient computer programming, for use as analysis and synthesis tools by the scientific and engineering community. These tools, for the most part, required the support of large central machines. In the meantime, slide rules were being replaced by hand-held calculators with trigonometric functions and which could be programmed for simple repetitive algorithms.

Today, large central processing systems are being replaced or at least supplemented with small, but powerful, mini and microcomputers. The development of the low cost microprocessor chip means that computers with capabilities that equal or exceed the earlier machines of the 1950s are now available in compact sizes. Sizes range from suitcase or desk top machines to file cabinet machines that can be expanded or configured to meet specialized needs. The microcomputer is becoming more and more affordable as a personal computing tool. The microcomputer or the "home computer" will be *the* engineering and scientific tool of this decade and the next! In addition, there is the widespread use of computer networking, making a wide variety of computer facilities dispersed around the country accessible to anyone with a microcomputer or terminal with an acoustic coupler and telephone, as well as a virtually limitless source of information.

*Microcomputer Tools for Communication Engineering* is intended to provide a set of invaluable tools for the design and operation of communication circuits. In many instances extensive, complicated computations are necessary for an enhanced design. This implies the use of large, sophisticated programs on large mainframe computer systems. In many problems the extensive capability of a sophisticated program is not required, where the communication circuit and its environment are not very complex, or the information sought only requires a simplified model. Viable alternatives include "stripped down" versions of sophisticated analysis tools that retain only the basic solution and the most frequently used options. These basic tools could be run on a mini or microcomputer. *Microcomputer Tools for Communication Engineering* is a set of these basic tools with the microcomputer in mind. The tools can also be implemented on mini or large computers that have the BASIC language capability. An attempt has been made to keep the programming straightforward with no machine dependent program statements. Thus the tools will be compatible with most BASIC languages.

*Microcomputer Tools for Communication Engineering* provides a series of automated capabilities (computer programs) directly oriented to communication engineering. These capabilities include the ability to analyze networks, filters, transmission lines, baluns, and antennas. A special capability is provided for the broadband matching of antennas. Radio operation will be enhanced through the use of the propagation and radio frequency interference tools.

A set of twenty-five programs are contained in this book. Each program is well-documented with theory, sample calculations, and computer listings. The programs are organized into the general areas of antennas, circuits, filters, impedance matching, propagation, radio frequency interference, and station operation. Chapters 1 through 5 deal with antenna design and analysis. Chapter 1 and Chapters 6 through 10 provide a series of automated circuit analysis tools. Chapters 11 through 13 give programs for filter design. Chapters 14 and 15 discuss impedance matching techniques. Chapters 16 through 19 are concerned with propagation and communication link analysis. Chapters 20 and 21 treat radio interference caused by intermodulation frequencies. Chapters 22 and 23 are related to satellite communications.

Chapters 24 and 25 provide practical tools and information for amateur radio operation. The latter chapter may be adapted to other frequency management applications. The appendix contains a description of the BASIC language used in the programs. This appendix will aid in any reprogramming necessary in the implementation on a particular computer, or aid in modifications to suit a special application.

# Contents

# Chapter 1

# Coupling




Mutual coupling betwen two monopoles in proximity to each other

## DESCRIPTION

This program calculates the input admittance, input impedance and coupling (power gain) for a two-port network with a given load. The program will also determine the load for maximum coupling. This analysis is useful for calculating coupling between antennas and coupling through an electric circuit.

## THEORY

When two or more antenna systems are in proximity to each other, power from a transmitting system can be coupled into the other systems. This problem can be analyzed using network "Y" parameters [1]. Any two antennas can be treated as a two-port network. In the figure below, terminals 1-1′ would be the feedpoint of the first antenna and terminals 2-2′ would be the feedpoint of the second antenna. By convention, the currents $I_1$ and $I_2$ are assumed into the network.

In general, of the four variables shown ($V_1, I_1, V_2, I_2$), only two are independent. Thus the following functions may be written

$$\begin{aligned} I_1 &= f_1(V_1, V_2) \\ I_2 &= f_2(V_1, V_2) \end{aligned} \qquad (1\text{-}1)$$


Two-port network definition

Assuming the functions are linear,

$$I_1 = Y_{11} V_1 + Y_{12} V_2$$
$$I_2 = Y_{21} V_1 + Y_{22} V_2 \qquad (1\text{-}2)$$

These admittance parameters are defined by

$$Y_{11} = \left.\frac{I_1}{V_1}\right| V_2 = 0 \qquad (1\text{-}3)$$

$$Y_{12} = \left.\frac{I_1}{V_2}\right| V_1 = 0 \qquad (1\text{-}4)$$

$$Y_{21} = \left.\frac{I_2}{V_1}\right| V_2 = 0 \qquad (1\text{-}5)$$

$$Y_{22} = \left.\frac{I_1}{V_2}\right| V_1 = 0 \qquad (1\text{-}6)$$

$V_1$ or $V_2$ equals zero implies that the terminals associated with these voltages are short circuited. An arrangement whereby these admittances might be computed (or measured) is shown below.




Arrangement for determining $Y_{11}$ and $Y_{21}$




Arrangement for determining $Y_{22}$ and $Y_{12}$

For the antenna coupling problem, the program in Chapter 4, MININEC, can be used to find the admittance parameters. The feedpoint of one antenna is excited and the feedpoint of the second antenna is short circuited. The calculated currents from MININEC are used with equations (1-3) and (1-5) to determine $Y_{11}$ and $Y_{21}$. In a similar manner $Y_{12}$ and $Y_{22}$ can be determined [2].

If coupling through a circuit is of interest, the program in Chapter 6, AC Circuit Analysis, would be used to determine the required admittance parameters.




Two-port network with a given load

Once the admittance parameters have been found, the input admittance and coupling can be calculated for arbitrary loads.

The output power is

$$P_0 = |V_2|^2 \, Re\,[Y_L] \qquad (1\text{-}7)$$

where $V_2$ is the output voltage (port 2), and $Re\,[Y_L]$ is the real part of the load admittance, $Y_L$.

The input power is

$$P_i = |V_1|^2 \, Re\,[Y_{in}] \qquad (1\text{-}8)$$

where $V_1$ is the input voltage (port 1), and $Re\,[Y_{in}]$ is the real part of the input admittance of port 1.

The power gain is then the ratio of (1-7) to (1-8):

$$G = \frac{P_0}{P_1} = \frac{|V_2|^2 \, Re\,[Y_L]}{|V_1|^2 \, Re\,[Y_{in}]} = \frac{Re\,[Y_L]}{Re\,[Y_{in}]} \left|\frac{V_2}{V_1}\right|^2 \qquad (1\text{-}9)$$

Notice that this ratio is dependent on the input admittance, $Y_{in}$, which can be calculated from equation (1-2) by solving $Y_{in} = I_1/V_1$ with $I_2 = -V_2 Y_L$. Then

$$Y_{in} = Y_{11} - \frac{Y_{21}\,Y_{12}}{Y_L + Y_{22}} \qquad (1\text{-}10)$$

Note that $Y_{in}$ depends on the load $Y_L$.

The two port admittance parameters in equation (1-2) can be used also to find the output to input voltage ratio, $V_2/V_1$, as follows:

$$\left|\frac{V_2}{V_1}\right| = \left|\frac{Y_{21}}{Y_L + Y_{22}}\right| \qquad (1\text{-}11)$$

Substituting (1-11) into (1-9), the gain becomes:

$$G = \frac{Re\,[Y_L]\,|Y_{21}|^2}{Re\,[Y_{in}]\,|Y_L + Y_{22}|^2} \qquad (1\text{-}12)$$

The gain (or coupling) in dB is

$$G_{dB} = 10 \log(G) \qquad (1\text{-}13)$$

see *Note* below

A $Y_L$ can be found that maximizes the gain. This gain is the maximum possible power transfer ratio *provided* that the generator is matched to the resulting $Y_{in}$ as given in (1-10).

---

Note: Throughout this book log(x) means logarithm of expression x with base 10, and ln(x) means natural logarithm of expression x, except in the program listings LOG(x) means natural logarithm of expression x as defined in the appendix for the BASIC language.

It is exceedingly difficult to maximize gain directly by taking the derivative of (1-12) with respect to a complex load admittance, $Y_L$, and setting it to zero. A more appropriate approach is the Linville Analysis used in the design of RF amplifiers [3,4]. Using this approach, the maximum coupling is then

$$G_{MAX} = \frac{[1-(1-L)^2]^{1/2}}{L} \qquad (1\text{-}14)$$

where

$$L = \frac{|Y_{12}\,Y_{21}|}{2\,Re[Y_{11}]Re[Y_{22}] - Re[Y_{12}\,Y_{21}]}$$

The matched load admittance on antenna 2 for maximum coupling is

$$Y_L = \left[\frac{1-\rho}{1+\rho} + 1\right] Re[Y_{22}] - Y_{22} \qquad (1\text{-}15)$$

where

$$\rho = \frac{G_{MAX}\,(Y_{12}\,Y_{21})^*}{|Y_{12}\,Y_{21}|}$$

and * indicates complex conjugate. The corresponding input admittance of port 1 (or antenna 1) is given in (1-10):

$$Y_{in} = Y_{11} - \frac{Y_{21}\,Y_{12}}{Y_L + Y_{22}} \qquad (1\text{-}16)$$

## REFERENCES


[1] M.E. Van Valkenberg, *Modern Network Synthesis*, New York: John Wiley and Sons, Inc., 1960.

[2] J.C. Logan, *et al.*, "Numerical Electromagnetic Code (NEC)," 1979 IEEE International Symposium on Electromagnetic Compatibility, San Diego, CA, October 1979.

[3] G. Johnson, "High Frequency Amplifier Using Admittance Parameters," *Electro Technology*, Dec. 1963, pp. 74-79.

[4] G. Johnson, "High Frequency Amplifier Using Admittance Parameters, Part 2," *Electro Technology*, Dec. 1963, pp. 66-73.


## PROGRAM NOTES

a. This program can be used with the program in Chapter 4, MININEC, to determine the coupling through antennas or with the program in Chapter 6, *AC Circuits Analysis*, for coupling between circuits.

b. All impedances are in ohms and admittances in mhos.

## SAMPLE PROBLEM

The two-port admittance parameters in this example have been determined using the MININEC program of Chapter 4. Given this information, determine the maximum coupling and the coupling for a 50 ohm load.

### COUPLING

```
THIS PROGRAM CALCULATES INPUT IMPEDANCE/ADMITTANCE
AND COUPLING FOR A TWO-PORT NETWORK.

Y11 (REAL,IMAGINARY)? .7738E-3,-.2022E-2
Y12 (REAL,IMAGINARY)? .4451E-3,-.1005E-3
Y21 (REAL,IMAGINARY)? .4451E-3,-.1005E-3
Y22 (REAL,IMAGINARY)? .7738E-3,-.2022E-2
```

```
TWO-PORT ADMITTANCE PARAMETERS
     Y11=  .7738E-03  +J -.002022
     Y12=  .4451E-03  +J -.1005E-03
     Y21=  .4451E-03  +J -.1005E-03
     Y22=  .7738E-03  +J -.002022

CHANGE TWO-PORT PARAMETERS (Y/N)? N
LOAD IMPEDANCE (REAL,IMAGINARY)
IF 0,0 IS ENTERED, ONLY MAXIMUM COUPLING IS COMPUTED? 50,0

     LOAD COUPLING
LOAD ADMITTANCE  =  .02           +J  0
LOAD IMPEDANCE   =  50            +J  0
INPUT ADMITTANCE =  .764419E-03   +J -.201861E-02
INPUT IMPEDANCE  =  164.07        +J  433.26
COUPLING (DB)    = -19.0292

NEW LOAD (Y/N)? Y
LOAD IMPEDANCE (REAL,IMAGINARY)
IF 0,0 IS ENTERED, ONLY MAXIMUM COUPLING IS COMPUTED? 0,0

     MAXIMUM COUPLING
LOAD ADMITTANCE  =  .638287E-03   +J  .196419E-02
LOAD IMPEDANCE   =  149.641       +J -460.488
INPUT ADMITTANCE =  .638287E-03   +J -.196419E-02
INPUT IMPEDANCE  =  149.641       +J  460.488
COUPLING (DB)    = -9.8194

NEW LOAD (Y/N)? N
```

## VARIABLE LIST

G = coupling

L1,L2 = load admittance or impedance

R1,I1 = $Y_{11}$

R2,I2 = $Y_{12}$

R3,I3 = $Y_{21}$

R4,I4 = $Y_{22}$

Y1,Y2 = input admittance or impedance

## PROGRAM LISTING

```
10 DIM B$(2)
20 B$(1)="MAXIMUM"
30 B$(2)="LOAD"
40 PRINT
50 PRINT "    "," COUPLING"
60 PRINT
70 PRINT "THIS PROGRAM CALCULATES INPUT IMPEDANCE/ADMITTANCE"
80 PRINT "AND COUPLING FOR A TWO-PORT NETWORK."
90 PRINT
100 PRINT "Y11 (REAL,IMAGINARY)";
110 INPUT R1,I1
```

```
120 PRINT "Y12 (REAL,IMAGINARY)";
130 INPUT R2,I2
140 PRINT "Y21 (REAL,IMAGINARY)";
150 INPUT R3,I3
160 PRINT "Y22 (REAL,IMAGINARY)";
170 INPUT R4,I4
180 N1=R2*R3-I2*I3
190 N2=R2*I3+R3*I2
200 PRINT
210 PRINT "TWO-PORT ADMITTANCE PARAMETERS"
220 PRINT "      Y11= ";R1;" +J "; I1
230 PRINT "      Y12= ";R2;" +J "; I2
240 PRINT "      Y21= ";R3;" +J "; I3
250 PRINT "      Y22= ";R4;" +J "; I4
260 PRINT
270 PRINT"CHANGE TWO-PORT PARAMETERS (Y/N)";
280 INPUT A$
290 IF A$="Y" GOTO 100
300 PRINT "LOAD IMPEDANCE (REAL,IMAGINARY)"
310 PRINT "IF 0,0 IS ENTERED, ONLY MAXIMUM COUPLING IS COMPUTED";
320 S=2
330 INPUT L1,L2
340 IF(L1+L2)=0 GOTO 680
350 D=L1*L1+L2*L2
360 L1=L1/D
370 L2=-L2/D
380 D1=L1+R4
390 D2=L2+I4
400 D=D1*D1+D2*D2
410 T1=(N1*D1+N2*D2)/D
420 T2=(D1*N2-D2*N1)/D
430 Y1=R1-T1
440 Y2=I1-T2
450 N=R3*R3+I3*I3
460 IF S=1 GOTO 480
470 G=(L1*N)/(D*Y1)
480 G=10*LOG(G)/LOG(10)
490 PRINT
500 PRINT "      ";B$(S);" COUPLING"
510 PRINT "LOAD ADMITTANCE  = ";L1;" +J ";L2
520 D=L1*L1+L2*L2
530 L1=L1/D
540 L2=-L2/D
550 PRINT "LOAD IMPEDANCE   = ";L1;" +J ";L2
560 PRINT "INPUT ADMITTANCE = ";Y1;" +J ";Y2
570 D=Y1*Y1+Y2*Y2
580 Y1=Y1/D
590 Y2=-Y2/D
600 PRINT "INPUT IMPEDANCE  = ";Y1;" +J ";Y2
610 PRINT "COUPLING (DB)    = ";G
620 IF S=0 GOTO 680
630 PRINT
640 PRINT "NEW LOAD (Y/N)";
650 INPUT A$
```

```
660 IF A$="N" GOTO 850
670 GOTO 300
680 N=SQR(N1*N1+N2*N2)
690 S=1
700 L=N/(2*R1*R4-N1)
710 IF L>S GOTO 830
720 IF L<0 GOTO 830
730 IF L<0.01 GOTO 760
740 G=(S-SQR(S-L*L))/L
750 GOTO 770
760 G=(L+.25*L*L*L)/2
770 T1=G*N1/N
780 T2=-G*N2/N
790 D=(S+T1)*(S+T1)+T2*T2
800 L1=R4*2*(S+T1)/D-R4
810 L2=-R4*2*T2/D-I4
820 GOTO 380
830 PRINT "COUPLING CANNOT BE COMPUTED FOR PARAMETERS"
840 GOTO 90
850 END
```